query: #defaultView:LineChart
  SELECT
  	?age ?count ?genderLabel
  WHERE
  {
  	{ SELECT ?gender ?age (count(*) as ?count) WHERE {
  		{ SELECT ?gender ?age WHERE {
    
           ?author p:P569/psv:P569 ?dob_node . # node for date of birth
           ?dob_node wikibase:timePrecision "11"^^xsd:integer . # date precision = day
           ?dob_node wikibase:timeValue ?dob . 
           ?author wdt:P21 ?gender .
           ?article wdt:P50 ?author .  
           ?article wdt:P31 wd:Q13442814 .
           ?article wdt:P577 ?date .
  
          FILTER(year(?date)>=2000)
  
          bind( 1000+year(?date)-year(?dob) as ?age_num )
          bind( SUBSTR(str(?age_num),1,4) as ?age )
    
          FILTER(BOUND(?age_num))
          FILTER(?age_num<=1100)
          FILTER(?age_num>=1000)
          }
           LIMIT 300000
          }
  	} GROUP BY ?gender ?age  }
  	SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
  }